

Training neural ODE mit Flux

Neurales ODE (ODE - einfache differentielle Gleichung) beschreibt eine Methode des Deep Learning in der neurale Netzwerke, dass die ODE, oder Teile davon abbildet.

Das Gegenstück aus der Werte eines ODE, sind Solver mit den einzelne Parameter des ODE angepasst werden. Zusätzlich existieren hybride Methoden,

Trainiert wird ein Multilayerperzeptron (MLP) mit den Schichten

1. 2->32, Aktivierungsfunktion tanh,
2. 32->2, Aktivierungsfunktion linear.

Eingabe in das Training: Zeitpunkte zu denen gemessene Temperaturen mit trainierten verglichen werden sollen.

Die zwei Knoten in den Eingabe- und Ausgabe-Layern entsprechen den Temperaturen an den Temperatursensoren des tclab, während die Heizquelle Nahe des ersten Temperatursensors zyklisch aufgeheizt wird. Das heißt, beide Sensoren werden zyklisch erhitzt. Sensor 1, stärker als Sensor 2. Sensor 2 ist im Aufheizen und Abkühlen im Vergleich zu Sensor 1 leicht verzögert.

Es soll ein neurales ODE mit dem Netzwerk angelernt werden. Das Netz soll die Änderungen des Verlauf des Temperaturen Vorhersagen.

Für die Vorhersage wird daher die Starttemperatur benötigt.

```
1 begin
2     using CSV
3     using DiffEqFlux
4     using Flux: train!, Chain, Dense
5     using Flux
6     using DataInterpolations
7     using DataFrames
8     using OrdinaryDiffEq
9     using Plots
10    using PlutoUI
11    using BenchmarkTools
12    using JLD2
13 end
```

Table of Contents

Training neural ODE mit Flux

Abspeichern des Models

Metriken

Änderungen und Fazit

Verwendung von zwei Temperaturen

Tolleranz vs Rechenzeit

Optimierer

Anzahl der Punkte

Anzahl der Heizzyklen

Optimierung des Aufgezeichneten Loss

Variation der Hiddenlayer und Anzahl der Knoten

Erhöhung der Anzahl der Epochen

celsius2kelvin (generic function with 1 method)

```
1 celsius2kelvin(c) = c + 273.15
```

prepare_dataframe (generic function with 1 method)

```
1 function prepare_dataframe(data_frame_path)
2
3     _df = DataFrame(CSV.File(data_frame_path))
4
5     start_times_stat = unique(_df, :run)
6
7     _df.start_time = start_times_stat[trunc.(Int, _df.run).+1, :time]
8     _df.duration = _df.time - _df.start_time
9     _df.T1 = _df.T1 .|> celsius2kelvin
10    _df.T2 = _df.T2 .|> celsius2kelvin
11    _df.Column1 = _df.Column1 .+ 1
12    _df.run = _df.run .+ 1
13
14    return _df
15 end
```

	Column1	T1	T2	Q1	time	run	start_time	duration
1	1	294.693	295.499	100.0	1.68496e9	1.0	1.68496e9	0.0
2	2	294.693	295.338	100.0	1.68496e9	1.0	1.68496e9	1.35462
3	3	294.693	295.466	100.0	1.68496e9	1.0	1.68496e9	2.5741
4	4	294.693	295.466	100.0	1.68496e9	1.0	1.68496e9	3.794
5	5	294.693	295.466	100.0	1.68496e9	1.0	1.68496e9	5.01308
6	6	294.693	295.563	100.0	1.68496e9	1.0	1.68496e9	6.23228
7	7	295.015	295.563	100.0	1.68496e9	1.0	1.68496e9	7.45149
8	8	295.015	295.531	100.0	1.68496e9	1.0	1.68496e9	8.67103
9	9	295.338	295.563	100.0	1.68496e9	1.0	1.68496e9	9.8905
10	10	295.338	295.531	100.0	1.68496e9	1.0	1.68496e9	11.1097
more								
8640	8640	298.238	299.527	0.0	1.68497e9	12.0	1.68496e9	868.37

```

1 begin
2     heating_df_path = joinpath(@__DIR__, "measurements_heating_and_cooling.csv")
3     heating_df = prepare_dataframe(heating_df_path)
4 end

```

[295.499, 295.338, 295.466, 295.466, 295.466, 295.563, 295.563, 295.531, 295.563, 295.

```

1 begin
2     # define data for the ODE
3
4     end_point = 3600;
5     tspan = (0.0, end_point);
6     tsteps = range(tspan[1], tspan[2], length=80);
7
8     # seconds since start
9     t = heating_df[1:end_point, :time] .- heating_df.start_time[1];
10    u = heating_df[1:end_point, :Q1];
11    T1 = heating_df[1:end_point, :T1];
12    T2 = heating_df[1:end_point, :T2];
13 end

```


```
([294.693, 305.386, 319.313, 329.699, 336.952, 342.39, 346.273, 349.113, 351.277, m
```

```
1 begin
2     # Interpoliere the u, y
3     u_t = ConstantInterpolation(u, t);
4     T1_t = LinearInterpolation(T1, t);
5     T2_t = LinearInterpolation(T2, t);
6     T0 = [T1_t(0), T2_t(0)]
7
8     ode_data = Array(T1_t(tsteps)), Array(T2_t(tsteps));
9 end
```


loss_n_ode (generic function with 1 method)

```
1 begin
2     # durch Interpolation langsam im Training
3     dudt = Flux.Chain(# x -> u_t.(x), # wie Verallgemeinerung auf andere u_t
4                     Flux.Dense(2=>32, tanh),
5                     Flux.Dense(32=>2)) |> f64
6
7     n_ode = NeuralODE(dudt, tspan, Tsit5(), saveat=tsteps, reltol=1e-3,
8                     abstol=1e-5)
9
10    y_predict_before = n_ode(T0)
11    ps = Flux.params(n_ode.p)
12
13    function predict_n_ode()
14        n_ode(T0)
15    end
16
17    loss_n_ode() = sum(abs2,ode_data[1] .- n_ode(T0)'[:, 1]) +
18                  sum(abs2,ode_data[2] .- n_ode(T0)'[:, 2])
19
20 end
```

```
1 begin
2
3      $\eta$  = 1e-4
4     opt_n_ode = ADAM( $\eta$ , (0.9, 0.8))
5
6     loss_vector = Vector{Float64}()
7     callback() = push!(loss_vector, loss_n_ode())
8
9     epochs1 = 500
10    data1 = Iterators.repeated(() , epochs1)
11
12    epochs2 = 4_500
13    data2 = Iterators.repeated(() , epochs2)
14
15    epochs3 = 45_000
16    data3 = Iterators.repeated(() , epochs3)
17
18    epochs = epochs1 + epochs2 + epochs3
19
20    Flux.train!(loss_n_ode, ps, data1, opt_n_ode, cb=callback)
21    opt_n_ode.eta = 5e-5
22    Flux.train!(loss_n_ode, ps, data2, opt_n_ode, cb=callback)
23    opt_n_ode.eta = 1e-5
24    Flux.train!(loss_n_ode, ps, data3, opt_n_ode, cb=callback)
25 end
```



```
1 plot(1:epochs, loss_vector .|> log10,  
2       label="nODE", title="Loss",  
3       xlabel = "Epochen", ylabel = "Loss log(mse)",)
```



```

1 begin
2     y_predict_after = n_ode(T0)
3
4     # plot result
5     p_node = plot(tsteps, ode_data[1], label="T1 °C", lc=:red, xlabel = "T /s",
6                 ylabel = raw"T /K", title="Heizkurven")
7     plot!(p_node, tsteps, ode_data[2], label="T2 °C", lc=:green)
8     scatter!(p_node, tsteps, y_predict_after[1, :], label="predicted T1 °C",
9             mc=:red)
10    scatter!(p_node, tsteps, y_predict_after[2, :], label="predicted T2 °C",
11            mc=:green)
12 end

```

Abspeichern des Models

```

1 begin
2     model_state = Flux.state(n_ode);
3     jldsave("nODE_Flux_model.jld2"; n_ode)
4 end

```

Metriken

```
1 using Statistics
```

r2 (generic function with 1 method)

```
1 r2(y_hat, y) = sum(abs2, y_hat .- mean(y)) / sum(abs2, y .- mean(y))
```

(313.225, 15.5394, 0.052984)

```
1 Flux.mse(y_predict_after[1,:], ode_data[1]),  
2 Flux.mae(y_predict_after[1,:], ode_data[1]),  
3 r2(y_predict_after[1,:], ode_data[1])
```

(59.8469, 6.76794, 0.139729)

```
1 Flux.mse(y_predict_after[2,:], ode_data[2]),  
2 Flux.mae(y_predict_after[2,:], ode_data[2]),  
3 r2(y_predict_after[2,:], ode_data[2])
```

Die Auswertung und der Vergleich erfolgt im Workbook: **nODE mit Lux**.

Änderungen und Fazit

Der Loss nimmt bis zu 1000 Zyklen ab und stagniert hiernach.

Die Oszilation der Temperaturen beim Heizen und Abkühlen konnte nicht nachgebildet werden.

Der Prozentsatz der Intensität der Heizung wird, testweise mit Interpolation als erste Layer hinzugefügt. Hierfür wurde eine ConstantInterpolation gewählt, da diese leicht schneller ist, als die Linear und bei den Wertebereich $[0, 100]$ geringere Fehler machen sollte.

Verwendung von zwei Temperaturen

Beim Training mit zwei Temperaturkurven ist der Verlauf des Losses glatter, als wenn nur ein Sensor angelernt wird.

Tolleranz vs Rechenzeit

Es wurden drei Parametersätze für die Auflösung des ODE getestet.

reltol	abstol	Rechenzeit pro Epoche \s
1e-3	1e-5	1.94
1e-5	1e-7	2.83
1e-7	1e-9	7.92

Der Verlauf Prognose ist zwischen den Parametersätzen nicht wirklich unterschiedlich. Zusätzlich wurde eine Lauf mit einer relativen Toleranz von $1e-2$ durchgeführt, hier hatte das System den Anschein, als ob zu schwingen beginnen würde. Allerdings kann es sich hier um ein Artefakt der hohen erlaubten Toleranz handeln, zusätzlich war der aufgezeichnete Loss sehr verrauscht, eine Optimierung gelang hier nicht. Der Test mit der Toleranz von $1e-3$ ein Optimum zu finden, wo das Rauschen verringt und dennoch Schwingung erhalten bleibt war nicht erfolgreich. Der Unterschied in den reichenzeit sind bei hoher und Mittlerer Toleranz nicht wesentlich im Vergleich zur niedrigen Toleranz erst hier werden anscheinend wesentlich mehr Zwischenschritte für die Berechnung benötigt. Die Rechenzeit bezoglich auf zwei Layer mit 32 versteckten Knoten mit einem AMD Ryzen 5800X bei 120 Messpunkten. Aktuell wird zur Beschleunigung mit 80 Punkten gerechnet.

Optimierer

Tsit5 gegen DP5 getauscht. Der Verlauf Prognose ist zwischen den beiden Optimierern nicht wesentlich unterschiedlich. Tsit5 war aber bei $reltol=1e-3$, $abstol=1e-5$ mit einer Rechenzeit von 2.5 s pro Epoche langsamer.

Anzahl der Punkte

Die Anzahl festgelegten Zwischenschritte (tsave) wurde erhöht, damit mehr Trainingsbeispiele

des Ausgabevektor des Netzes mit (interpolierten) Werten verglichen werden können, um so die Anzahl der Beispiele zu erhöhen. Die Anzahl wurde erst verdoppelt und dann verdreifacht. Dies führe jedoch nicht zu besseren Optimierungen. Daher wurde die Anzahl wieder auf Aktuell 80 Punkte reduziert.

Anzahl der Heizzyklen

Desweiteren wurde getestet, ob mehr Heizzyklen als Beispiel in das Training überführt werden können. Dies führte zu keiner verbesserung der Optimierung.

Optimierung des Aufgezeichneten Loss

Die Funktion aufgezeichnete Loss war zum Teil sehr rauch. Es wurde vermutet, dieses dies insbesondere, dass weitere Lernen verhindert. Deswegen wurden unterschiedliche Epochenanzahlen mit unterschiedlichen Lernraten getestet. Niedrige Lernraten führten zwar zur klatteren Kurven, aber nicht zur besseren Optimierungen.

Variation der Hiddenlayer und Anzahl der Knoten

Es wurde regelmäßig die Anzahl Layer von zwei bis zu fünf getestet. Gleichzeitig wurde die Anzahl der Knoten von 16, 32, 64 geprüft. Beide Veränderungen führen zu kleiner wesentlichen Verbesserung im Loss. Das Modell fängt nicht an zu schwingen.

Erhöhung der Anzahl der Epochen

Durch all die Optimierungen der Rechenzeit und der Reduzierung des Rauschens im Loss gelang, dass bei der Erhöhung der Anzahl der Epochen ein geringer Fortschritt beim Lernen möglich wurde. Nun gelingt es durch die Erhöhung der Epochen von 1000 auf 50000 für die erste Heizphase eine Abweichung von der Mitteltemperatur zu erreichen.